# ODROID-C1 Amlogic libraries
# Maintainer: Kevin Mihelich <kevin@archlinuxarm.org>
# Contributor: Jan Holthuis <holthuis.jan@googlemail.com>

buildarch=4

pkgname=aml-libs-c1
pkgver=r11.820b12a
pkgrel=5
arch=('armv7h')
pkgdesc="ODROID-C1 Amlogic Libraries"
url="http://www.hardkernel.com/"
license=('Proprietary')
depends=('alsa-lib')
makedepends=('git')
source=("git+https://github.com/mdrjr/c1_aml_libs.git"
        '99-amlogic.rules')
md5sums=('SKIP'
         'c8f7f6b83416768cc9424f7e7a372fa4')

pkgver() {
  cd c1_aml_libs
  printf "r%s.%s" "$(git rev-list --count HEAD)" "$(git rev-parse --short HEAD)"
}

build() {
  cd c1_aml_libs
  make clean
  make
}

package() {
  cd c1_aml_libs
  install -d "${pkgdir}"/usr/include/{,amports,cutils,ppmgr,amcodec,amadec,amavutils}
  install -d "${pkgdir}/etc/ld.so.conf.d"
  install -d "${pkgdir}/usr/lib/aml_libs"

  install -m644 aml.conf "${pkgdir}/etc/ld.so.conf.d"

  install -m755 amadec/libamadec.so amavutils/libamavutils.so amcodec/libamcodec.so "${pkgdir}/usr/lib/aml_libs"
  install -m644 amadec/{,include/}*.h "${pkgdir}/usr/include/amadec"
  install -m644 amavutils/include/*.h "${pkgdir}/usr/include/amavutils"
  install -m644 amavutils/include/cutils/*.h "${pkgdir}/usr/include/cutils"
  install -m644 amcodec/include/*.h "${pkgdir}/usr/include/amcodec"
  install -m644 amcodec/include/*.h "${pkgdir}/usr/include"
  install -m644 amcodec/include/amports/*.h "${pkgdir}/usr/include/amports"
  install -m644 amcodec/include/ppmgr/*.h "${pkgdir}/usr/include/ppmgr"

  # install /dev/am* permission fix
  install -Dm644 "${srcdir}/99-amlogic.rules" "${pkgdir}/usr/lib/udev/rules.d/99-amlogic.rules"
}
